

SERIAL PRINTER MODULE

Order Code


BAS120	Serial Printer Module
FRM015	Serial Printer Firmware Chip

Serial Printer Module Contents

- 1 x Product Information Sheet
- 1 x Serial Printer Module
- 4 x Self Adhesive Feet

Also required:

Standard Centronics printer cable.



Assembly

Stick the 4 self adhesive feet to the bottom of the Serial Printer Module.

Introduction

The serial printer module allows the Stamp to print directly via the serout command, without the need for a computer. This is primarily used for datalogging or monitoring results. It is generally envisaged that the module will be used with older dot-matrix type printers (old BBC printers are ideal!), but it can also be used with modern inkjets and laser printers.

It is important that the printer is switched off when the cables, modules etc. are connected together. If not it is possible for random characters to be printed due to noise spikes as the modules are connected. In practice we would recommend that the printer is only switched on directly before the Stamp program is downloaded, and switched off again after the program has run.

Note that on dot-matrix (and some ink-jet) printers each line is printed separately when the carriage return (CR-13) character is sent. On most inkjets and all laser printers the whole page is printed when the form feed (FF-12) character is sent.

Connections

The Serial Printer Module requires three connections to the Stamp: V+ (5V), Ground (0V), and serial pin. When using the Stamp Controller these connections are automatically made. Serial communication is via pin 7, and that is the pin used in all the example programs. Use the standard printer cable to connect the Serial Printer Module to the printer. **WARNING: Always switch the printer OFF when making connections!**

Limitations

The Serial Printer Module provides a basic text printing facility, using the default printer font (the font and font size cannot be altered). It does not support all the 'intelligent' features of modern printers (e.g. graphics). In particular the error-reporting is very basic, and so it is important to ensure the printer is correctly set-up (i.e. paper is present etc.) before trying to print.

Operation:

The Serial Printer Module must be 'woken-up' with a start of text (STX-2) command before printing, and should be disabled with an end of text (ETX-3) command after printing is over. All lines must be followed by a carriage return (CR-13) and line feed (LF-10) sequence. All pages must be terminated with a form feed (FF-12) character. The following Basic Stamp program illustrates how printing is used with the serout command.

```

`PROGRAM: PRT001.BAS

`BASIC Stamp program that demonstrates
`how to use the serial Printer Module.

`Connect the Printer Module to the Stamp.
`Switch the printer OFF, connect the printer
`cable to the module. Power up the Stamp,
`and then switch the printer ON. Then download
`this program.

symbol      S_OUT =    7      `serial Printer output (output)
symbol      i      =   b5     `scratchpad counter

`Define useful ASCII characters
symbol      STX    =    2     `Start of text ("wake-up")
symbol      ETX    =    3     `End of text ("sleep")
symbol      TAB    =    9     `Tab
symbol      LF     =   10     `Line Feed
symbol      FF     =   12     `Form Feed

`Note that carriage return CR = 13 is a pre-defined constant
`within the Stamp Editor and therefore does not require a
`separate `symbol' definition.

init: `Make serial pin an output, and leave high for 5ms
      `This provides time for the printer module serial
      `input to reset.

      high S_OUT          `Make pin7 high output.
      pause 5             `Short pause.

main: `Wake up printer module via sending STX command.
      serout S_OUT,T2400,(STX)

      `Print `Hello' message.
      serout S_OUT,T2400,("Hello",CR,LF)

      `Note all lines must always be followed by CR then LF.

      `Print ten more lines.
      for i = 1 to 10
        serout S_OUT,T2400,(TAB,"This is line ",#i,CR,LF)
      next i

      `Note the # before the variable i. This ensures
      `variable numbers are transmitted as a string of
      `ASCII characters rather than as a single byte.

      `Note also some printers print very near left edge.
      `Printing a TAB before each line helps create a margin.

      `Now finish printing by sending form-feed character.
      serout S_OUT,T2400,(FF)

      `Finally send printer driver module to sleep.
      serout S_OUT,T2400,(ETX)

      end

```

Basic Stamp and PBasic are registered trademarks of Parallax, Inc.